
Change 1 

	Function Name
	a_IP_HTTP_Handler()

	Reason for change
	1)  If UE attached for both IPv6 and IPv4, UE can use IPv6 for the first PDN and IPv4 for second PDN. Currently, TTCN configures ipv6 for second PDN if ipv6 used for first PDN. Hence, changes needed to add ipv4 or ipv6 based on ipv4 or ipv6 are used for first PDN.
2)  As UE can use either IPv4 or IPv6, hence, connection request received with either IP address and stored the connection details later.

	Summary of change
	1) Configured IPv4 or IPv6 if earlier IPv6 or IPv4 used for the first PDN using f_IpAddressIsIPv4() and f_IpAddressIsIPv6()
2) Server address in accepting the connection is any. Later, conneciton details stored based on received local Ipaddress.

	Source of change
	Common\IP_PTC\IP_PTC_HTTP_Handler.ttcn

	MCC160 comments
	


Before change:

	altstep a_IP_HTTP_Handler(inout HttpInfo_Type p_HttpInfo) runs on IP_PTC

  {

    var HTTP_CTRL_REQ v_HTTP_CTRL_REQ;

    var HTTP_DATA_REQ v_HTTP_DATA_REQ;

    var IP_SOCKET_IND v_IP_SOCKET_IND;

    var template (value) IP_AddrInfo_Type v_ServerAddr;

    var PortNumber_Type v_PortNumber;

    var template (omit) IP_Connection_Type v_XcapConnectionByRef;

    var HttpRequest_Type v_HttpRequest;

    var HttpResponse_Type v_HttpResponse;

    var TCP_Data_Type v_Data;

    var bitstring v_Bitstring;

    var boolean v_XcapServerIsStarted := ispresent(p_HttpInfo.XcapServer);

    var boolean v_XcapServerIsConnected := ispresent(p_HttpInfo.XcapServer.Connection);

    [] HTTPCTRL.receive(cr_HttpCtlReq) -> value v_HTTP_CTRL_REQ

      {

        if (isvalue(v_HTTP_CTRL_REQ.authenticationMechanism)) {

          p_HttpInfo.HttpAuthenticationMechanism := valueof(v_HTTP_CTRL_REQ.authenticationMechanism);

        }

        if (ispresent(v_HTTP_CTRL_REQ.xcapServer)) {

          if (v_XcapServerIsStarted) {

            f_TcpServer_CloseConnection(p_HttpInfo.XcapServer.Connection);             // close IP connection (if any)

            f_TcpServer_Stop(p_HttpInfo.XcapServer.ServerAddr, p_HttpInfo.XcapServer.PortNumber); // stop HTTP server

          }

          select (p_HttpInfo.HttpAuthenticationMechanism) {

            case (noAuthentication, httpDigestAuthentication) {

              v_ServerAddr := f_Char2IpAddress(v_HTTP_CTRL_REQ.xcapServer.serverAddr);

              v_PortNumber := v_HTTP_CTRL_REQ.xcapServer.serverPort;

              p_HttpInfo.XcapServer := valueof(cs_HttpServer_Init(v_ServerAddr, v_PortNumber));

              f_TcpServer_Start(v_ServerAddr, v_PortNumber, v_HTTP_CTRL_REQ.drbInfo, http, v_HTTP_CTRL_REQ.tlsInfo);

            }

            case (gaaAuthentication) {

              FatalError(__FILE__, __LINE__, "gaaAuthentication is not supported by IP PTC (yet)");

            }

            case else {

              FatalError(__FILE__, __LINE__, "invalid HttpAuthenticationMechanism_Type");

            }

          }

        }

        if (ispresent(v_HTTP_CTRL_REQ.bsfServer)) {

          // FFS

        }

        HTTPCTRL.send(cs_HTTP_CTRL_CNF);

      }

    [v_XcapServerIsStarted] a_TcpServer_AcceptConnection(v_XcapConnectionByRef, cr_IP_Socket(p_HttpInfo.XcapServer.ServerAddr, p_HttpInfo.XcapServer.PortNumber))
      {

        if (isvalue(p_HttpInfo.XcapServer.Connection)) {     /* @sic R5-145732 sic@ */

          f_TcpServer_CloseConnection(p_HttpInfo.XcapServer.Connection, false);       // if there is already a connection => close the old connection; WaitForCNF==false

          f_ErrorLog(__FILE__, __LINE__, "XCAP server connection already exists - closed");

        }

        p_HttpInfo.XcapServer.Connection := valueof(v_XcapConnectionByRef);   // a_TcpServer_AcceptConnection 'returns' a valid connection i.e. v_ServerConnectionByRef will never be omit

      }

    [v_XcapServerIsConnected] a_TcpServer_CloseConnection(p_HttpInfo.XcapServer.Connection)

      {

        p_HttpInfo.XcapServer.Connection := omit;

      }

    [v_XcapServerIsConnected] HTTPDATA.receive(cr_DataReq) -> value v_HTTP_DATA_REQ

      {

        v_HttpResponse := v_HTTP_DATA_REQ.httpResponse;

        v_Bitstring := encvalue(v_HttpResponse);

        v_Data := bit2oct(v_Bitstring);

        IP_SOCK.send(cs_TCP_DATA_REQ(p_HttpInfo.XcapServer.Connection, v_Data));

      }

    [v_XcapServerIsConnected] IP_SOCK.receive(cr_TCP_DATA_IND(p_HttpInfo.XcapServer.Connection)) -> value v_IP_SOCKET_IND

      {

        v_Data := v_IP_SOCKET_IND.DATA.Ind.TCP.Recv;

        v_Bitstring := oct2bit(v_Data);

        if (decvalue(v_Bitstring, v_HttpRequest) != 0) {

          FatalError(__FILE__, __LINE__, "HTTP Request cannot be decoded");

        } else {

          HTTPDATA.send(cs_HTTP_DATA_IND(f_IP_AddrInfo_AddrStr(p_HttpInfo.XcapServer.ServerAddr), p_HttpInfo.XcapServer.PortNumber, v_HttpRequest));

        }

      }

  }


After change :

	   altstep a_IP_HTTP_Handler(inout HttpInfo_Type p_HttpInfo) runs on IP_PTC

  {

    var HTTP_CTRL_REQ v_HTTP_CTRL_REQ;

    var HTTP_DATA_REQ v_HTTP_DATA_REQ;

    var IP_SOCKET_IND v_IP_SOCKET_IND;

    var template (value) IP_AddrInfo_Type v_ServerAddr;

    var PortNumber_Type v_PortNumber;

    var template (omit) IP_Connection_Type v_XcapConnectionByRef;

    var HttpRequest_Type v_HttpRequest;

    var HttpResponse_Type v_HttpResponse;

    var TCP_Data_Type v_Data;

    var bitstring v_Bitstring;

    var boolean v_XcapServerIsStarted := ispresent(p_HttpInfo.XcapServer);

    var boolean v_XcapServerIsConnected := ispresent(p_HttpInfo.XcapServer.Connection);

    [] HTTPCTRL.receive(cr_HttpCtlReq) -> value v_HTTP_CTRL_REQ

      {

        if (isvalue(v_HTTP_CTRL_REQ.authenticationMechanism)) {

          p_HttpInfo.HttpAuthenticationMechanism := valueof(v_HTTP_CTRL_REQ.authenticationMechanism);

        }

        if (ispresent(v_HTTP_CTRL_REQ.xcapServer)) {

          if (v_XcapServerIsStarted) {

            f_TcpServer_CloseConnection(p_HttpInfo.XcapServer.Connection);             // close IP connection (if any)

            f_TcpServer_Stop(p_HttpInfo.XcapServer.ServerAddr, p_HttpInfo.XcapServer.PortNumber); // stop HTTP server

          }

          select (p_HttpInfo.HttpAuthenticationMechanism) {

            case (noAuthentication, httpDigestAuthentication) {

              v_ServerAddr := f_Char2IpAddress(v_HTTP_CTRL_REQ.xcapServer.serverAddr);

              v_PortNumber := v_HTTP_CTRL_REQ.xcapServer.serverPort;

              p_HttpInfo.XcapServer := valueof(cs_HttpServer_Init(v_ServerAddr, v_PortNumber));

              f_TcpServer_Start(v_ServerAddr, v_PortNumber, v_HTTP_CTRL_REQ.drbInfo, http, v_HTTP_CTRL_REQ.tlsInfo);

              //tmp NK XCAP ipv4 started with IPv6

              if( not f_IpAddressIsIPv4(v_HTTP_CTRL_REQ.xcapServer.serverAddr)) {

               f_TcpServer_Start(cs_IPv4Addr(px_IPv4_Address2_NW), v_PortNumber, v_HTTP_CTRL_REQ.drbInfo, http, v_HTTP_CTRL_REQ.tlsInfo);

              }

              else if(not f_IpAddressIsIPv6(v_HTTP_CTRL_REQ.xcapServer.serverAddr)){

               f_TcpServer_Start(cs_IPv6Addr(px_IPv6_Address2_NW), v_PortNumber, v_HTTP_CTRL_REQ.drbInfo, http, v_HTTP_CTRL_REQ.tlsInfo);

              }
            }

            case (gaaAuthentication) {

              FatalError(__FILE__, __LINE__, "gaaAuthentication is not supported by IP PTC (yet)");

            }

            case else {

              FatalError(__FILE__, __LINE__, "invalid HttpAuthenticationMechanism_Type");

            }

          }

        }

        if (ispresent(v_HTTP_CTRL_REQ.bsfServer)) {

          // FFS

        }

        HTTPCTRL.send(cs_HTTP_CTRL_CNF);

      }

    //[v_XcapServerIsStarted] a_TcpServer_AcceptConnection(v_XcapConnectionByRef, cr_IP_Socket(p_HttpInfo.XcapServer.ServerAddr, p_HttpInfo.XcapServer.PortNumber))

    [v_XcapServerIsStarted] a_TcpServer_AcceptConnection(v_XcapConnectionByRef, cr_IP_Socket(?, p_HttpInfo.XcapServer.PortNumber)) 
      {

        if (isvalue(p_HttpInfo.XcapServer.Connection)) {     /* @sic R5-145732 sic@ */

          f_TcpServer_CloseConnection(p_HttpInfo.XcapServer.Connection, false);       // if there is already a connection => close the old connection; WaitForCNF==false

          f_ErrorLog(__FILE__, __LINE__, "XCAP server connection already exists - closed");

        }

        p_HttpInfo.XcapServer.Connection := valueof(v_XcapConnectionByRef);   // a_TcpServer_AcceptConnection 'returns' a valid connection i.e. v_ServerConnectionByRef will never be omit

        p_HttpInfo.XcapServer.ServerAddr := p_HttpInfo.XcapServer.Connection.Local.IpAddr;
      }

    [v_XcapServerIsConnected] a_TcpServer_CloseConnection(p_HttpInfo.XcapServer.Connection)

      {

        p_HttpInfo.XcapServer.Connection := omit;

      }

    [v_XcapServerIsConnected] HTTPDATA.receive(cr_DataReq) -> value v_HTTP_DATA_REQ

      {

        v_HttpResponse := v_HTTP_DATA_REQ.httpResponse;

        v_Bitstring := encvalue(v_HttpResponse);

        v_Data := bit2oct(v_Bitstring);

        IP_SOCK.send(cs_TCP_DATA_REQ(p_HttpInfo.XcapServer.Connection, v_Data));

      }

    [v_XcapServerIsConnected] IP_SOCK.receive(cr_TCP_DATA_IND(p_HttpInfo.XcapServer.Connection)) -> value v_IP_SOCKET_IND

      {

        v_Data := v_IP_SOCKET_IND.DATA.Ind.TCP.Recv;

        v_Bitstring := oct2bit(v_Data);

        if (decvalue(v_Bitstring, v_HttpRequest) != 0) {

          FatalError(__FILE__, __LINE__, "HTTP Request cannot be decoded");

        } else {

          HTTPDATA.send(cs_HTTP_DATA_IND(f_IP_AddrInfo_AddrStr(p_HttpInfo.XcapServer.ServerAddr), p_HttpInfo.XcapServer.PortNumber, v_HttpRequest));

        }

      }

  }


