
Change 1 
	Function name
	fl_SPIs_Init

	Reason for change
	According to the TTCN language specification, Appendix B.1.2.8, templates cannot be set to omit at runtime, except when applied to optional fields of record and set templates. 
fl_SPIs_Init uses two integer templates p_PrevSPI_ps and p_PrevSPI_pc which actually take value omit upon initial registration. A compile and runtime system implementing above restriction will abort then at runtime. 

	Summary of change
	Re-wrote  fl_SPIs_Init such that the matching operation is only performed when these templates have a value. In case they have no value, it is safe to assign the SPI.

	TTCN module
	Common/IMS/IMS_Component

	MCC160 Comment
	


Before change

   function fl_SPIs_Init(integer p_NewSPI_us,

                        integer p_NewSPI_uc,

                        template (omit) integer p_PrevSPI_ps := omit,

                        template (omit) integer p_PrevSPI_pc := omit) runs on IMS_PTC return IMS_SPIs_Type

  { /* @sic R5s130266 change 1.3, 4, 5 - MCC160 Implementation: taking into account the previous SPIs of the network side sic@ */

    /* p_PrevSPI_ps, p_PrevSPI_pc are omit at initial registration */

    var IMS_SPIs_Type v_SPIs := valueof(cs_IMS_SPIs_Init(p_NewSPI_us, p_NewSPI_uc));

    var integer v_Index := enum2int(vc_IMS_Global.PdnIndex);

    var integer v_SPI := 256 + (v_Index * 1024);  /* According to RFC 2406, SPI values from 0 to 255 are reserved

                                                     @sic R5s130651: v_Index used to get different SPIs for different PDNs sic@ */

    if (p_NewSPI_uc == p_NewSPI_us) {

      FatalError(__FILE__, __LINE__, "SPIs of UE side have the same values");

    }

    while (not isvalue(v_SPIs.SPI_ps) or not isvalue(v_SPIs.SPI_pc)) {

      if ((v_SPI != p_NewSPI_uc) and (v_SPI != p_NewSPI_us) and                // new value shall be neither equal to the UE's values ...

          not match(v_SPI, p_PrevSPI_pc) and not match(v_SPI, p_PrevSPI_ps)) { // ... nor to the network's previous values

        if (not isvalue(v_SPIs.SPI_ps)) {

          v_SPIs.SPI_ps := v_SPI;

        } else {

          v_SPIs.SPI_pc := v_SPI;

        }

      }

      v_SPI := v_SPI + 1;

    }

    return v_SPIs;

  }     
After change

  function fl_SPIs_Init(integer p_NewSPI_us,

                        integer p_NewSPI_uc,

                        template (omit) integer p_PrevSPI_ps := omit,

                        template (omit) integer p_PrevSPI_pc := omit) runs on IMS_PTC return IMS_SPIs_Type

  { /* @sic R5s130266 change 1.3, 4, 5 - MCC160 Implementation: taking into account the previous SPIs of the network side sic@ */

    /* p_PrevSPI_ps, p_PrevSPI_pc are omit at initial registration */

    var IMS_SPIs_Type v_SPIs := valueof(cs_IMS_SPIs_Init(p_NewSPI_us, p_NewSPI_uc));

    var integer v_Index := enum2int(vc_IMS_Global.PdnIndex);

    var integer v_SPI := 256 + (v_Index * 1024);  /* According to RFC 2406, SPI values from 0 to 255 are reserved

                                                     @sic R5s130651: v_Index used to get different SPIs for different PDNs sic@ */

    if (p_NewSPI_uc == p_NewSPI_us) {

      FatalError(__FILE__, __LINE__, "SPIs of UE side have the same values");

    }

    while (not isvalue(v_SPIs.SPI_ps) or not isvalue(v_SPIs.SPI_pc)) {

      if ((v_SPI != p_NewSPI_uc) and (v_SPI != p_NewSPI_us) and                // new value shall be neither equal to the UE's values ...



((not isvalue(p_PrevSPI_pc)) or (isvalue(p_PrevSPI_pc) and not match(v_SPI, p_PrevSPI_pc))) and 


    ((not isvalue(p_PrevSPI_ps)) or (isvalue(p_PrevSPI_ps) and not match(v_SPI, p_PrevSPI_ps)))) { //

          if (not isvalue(v_SPIs.SPI_ps)) {

            v_SPIs.SPI_ps := v_SPI;

          } else {

            v_SPIs.SPI_pc := v_SPI;

          }

      }

      v_SPI := v_SPI + 1;

    }

    return v_SPIs;

  }     
